

Using CPAN's Toolchain to Improve Your Code

Tom Hukins

More Than One Way..

- CPAN
 - Modular and Reusable
 - Tested
 - Documented
- Other
 - ...

```
#!perl

# Get information about YAPC::Europe 2009 in Lisbon

use strict;
use warnings;

use 5.010;

use Getopt::Long 'GetOptions';
use HTML::Entities 'decode_entities';
use LWP::Simple 'get';

my %opt;
GetOptions (\%opt, qw(conference talk weather));
```

```
if ($opt{conference}) {  
    my $html = get 'http://yapceurope2009.org/ye2009/stats';  
    $html =~ m/(\d+) committed users/;  
    say "This conference has $1 attendees";  
}  
  
if ($opt{talk}) {  
    my $html = get 'http://yapceurope2009.org/ye2009/talk/2053';  
    my $count = () = $html =~ m{<li><a href="/ye2009/user">}g;  
    say "There are $count people in this room. Hello!";  
}  
  
if ($opt{weather}) {  
    my $html = get 'http://news.bbc.co.uk/weather/forecast/48';  
    $html =~ m{<span class="cent">([^\<]+)</span>};  
    my $temperature = decode_entities($1);  
    say "The temperature in Lisbon is $temperature";  
}
```

Run the Code

```
% perl yapc.pl -conference  
This conference has 325 attendees
```

Write a Module

```
package This::Conference;

use strict;
use warnings;

use 5.010;

our $VERSION = '1.23';

use HTML::Entities 'decode_entities';
use LWP::Simple 'get';

sub attendees {
    my $html = get 'http://yapceurope2009.org/ye2009/stats';
    $html =~ m/(\d+) committed users/;

    return $1;
}
```

Write a Module

```
sub attendees_at_talk {
    my $class = shift;
    my $talk_id = shift;

    my $html = get 'http://yapceurope2009.org/ye2009/talk/2053';
    my $count = () = $html =~ m{<li><a href="/ye2009/user">}g;

    return $count;
}

sub temperature {
    my $html = get 'http://news.bbc.co.uk/weather/forecast/48';
    $html =~ m{<span class="cent">([^\<]+)</span>};
    my $temperature = decode_entities($1);

    return $temperature;
}
```

Write a Script

```
#!perl

use strict;
use warnings;

use 5.010;

use Getopt::Long 'GetOptions';
use This::Conference ();

my %opt;
GetOptions(\%opt, qw(conference talk weather));
```

Write a Script

```
if ($opt{conference}) {  
    my $count = This::Conference->attendees();  
    say "This conference has $count attendees";  
}  
  
if ($opt{talk}) {  
    my $count = This::Conference->attendees_at_talk(2053);  
    say "There are $count people in this room. Hello!";  
}  
  
if ($opt{weather}) {  
    my $temperature = This::Conference->temperature();  
    say "The temperature in Lisbon is $temperature";  
}
```

Write a Makefile.PL

```
use strict;
use warnings;
use ExtUtils::MakeMaker;
WriteMakefile(
    NAME => 'This::Conference',
    MIN_PERL_VERSION => 5.010000,
    VERSION_FROM => 'lib/This/Conference.pm',
    ABSTRACT => 'YAPC::Europe 2009 Information',
    AUTHOR => 'Tom Hukins <tom@eborcom.com>',
    PREREQ_PM => {
        'Getopt::Long' => 0,
        'HTML::Entities' => 0,
        'LWP::Simple' => 0,
    },
    EXE_FILES => ['bin/yapc2009'],
) ;
```

Document Your Module

```
=head1 NAME
```

```
This::Conference - Information about YAPC::Europe 2009
```

```
=head1 SYNOPSIS
```

```
Find out about YAPC::Europe 2009. Provides the following methods:
```

```
=over
```

```
=item * attendees
```

```
=item * attendees_at_talk
```

```
=item * temperature
```

```
=back
```

```
=head1 SEE ALSO
```

```
L<yapc2009>, http://yapceurope2009.org/.
```

```
=cut
```

Not Like This!

```
###  
# This::Conference  
###  
#  
# Information about YAPC::Europe 2009  
#  
# This was written by Tom Hukins for YAPC::Europe  
# on some date that you don't care about.  
#  
# Your version control system should tell you  
# that anyway.  
#  
# So this documentation tells you nothing useful.  
#  
###
```

Write Some Tests

```
# !perl  
  
use strict;  
use warnings;  
  
use Test::More tests => 1;  
  
use_ok 'This::Conference';
```

Check Everything Works

```
% perl Makefile.PL && make test
Checking if your kit is complete...
Warning: the following files are missing in your kit:
        MANIFEST
Please inform the author.
Writing Makefile for This::Conference
cp lib/This/Conference.pm blib/lib/This/Conference.pm
cp bin/yapc2009 blib/script/yapc2009
/usr/bin/perl "-MExtUtils::MY" -e "MY->fixin(shift)"
    blib/script/yapc2009
PERL_DL_NONLAZY=1 /usr/bin/perl "-MExtUtils::Command::MM" "-e"
    "test_harness(0, 'blib/lib', 'blib/arch')" t/*
t/compile....ok
All tests successful.
Files=1, Tests=1,  0 wallclock secs ( 0.12 cusr +  0.02 csys =
 0.14 CPU)
```

Build a MANIFEST

```
% make manifest  
/usr/bin/perl "-MExtUtils::Manifest=mkmanifest"  
    -e mkmanifest  
  
% cat MANIFEST  
bin/yapc2009  
lib/This/Conference.pm  
Makefile.PL  
MANIFEST      This list of files  
t/compile.t
```

Build the Distribution

```
make dist
rm -rf This-Conference-1.23
/usr/bin/perl "-MExtUtils::Manifest=manicopy,maniread" \
    -e "manicopy(maniread(), 'This-Conference-1.23', 'best');"
mkdir This-Conference-1.23
mkdir This-Conference-1.23/t
mkdir This-Conference-1.23/bin
mkdir This-Conference-1.23/lib
mkdir This-Conference-1.23/lib/This
Generating META.yml
tar cvf This-Conference-1.23.tar This-Conference-1.23
This-Conference-1.23/
This-Conference-1.23/MANIFEST
This-Conference-1.23/t/
This-Conference-1.23/t/compile.t
This-Conference-1.23/bin/
This-Conference-1.23/bin/yapc2009
This-Conference-1.23/lib/
This-Conference-1.23/lib/This/
This-Conference-1.23/lib/This/Conference.pm
This-Conference-1.23/Makefile.PL
This-Conference-1.23/META.yml
rm -rf This-Conference-1.23
gzip --best This-Conference-1.23.tar
```

Helpful Tools

- h2xs
- Module::Starter
- Module::Setup
- Dist::Zilla
- Module::Install

Test Coverage Reports

```
% PERL5OPT=-MDevel::Cover make test
cp lib/This/Conference.pm blib/lib/This/Conference.pm
cp bin/yapc2009 blib/script/yapc2009
/home/tom/perl-inst/perl-5.10/bin/perl -MExtUtils::MY
-e 'MY->fixin(shift)' -- blib/script/yapc2009
PERL_DL_NONLAZY=1 /home/tom/perl-inst/perl-5.10/bin/perl
"-MExtUtils::Command::MM" "-e" "test_harness(0, 'blib/lib',
'blib/arch')" t/*
t/compile.t .. ok
All tests successful.
Files=1, Tests=1, 98 wallclock secs ( 0.09 usr 0.02 sys +
91.89 cusr 0.57 csys = 92.57 CPU)
Result: PASS
```

Test Coverage Reports

```
% cover -report html_basic  
Reading database from /home/tom/sdhc/cpan-toolchain-talk/This-Conference/cover_db
```

File	stmt	bran	cond	sub	pod	time	total
blib/lib/This/Conference.pm	60.0	n/a	n/a	66.7	100.0	100.0	64.3
Total	60.0	n/a	n/a	66.7	100.0	100.0	64.3

Perldoc::Server

CPAN::Mini

```
local: /path/to/cpan-mini  
remote: http://cpan-mirror.example.org/
```

CPAN::Mini

% minicpan

CPAN::Mini::Inject

```
% mcpansi -add \
--authorid TOMHUKINS \
--all-in-meta \
--file This-Conference-1.23.tar.gz

% mcpansi --inject
```

CPAN::Mini::Webserver

The End!

tom@eborcom.com